

# F15JWatch Launcher

by [micono@mac.com](mailto:micono@mac.com) CC BY

このプログラムは、笹野潔氏が作られた、F15JWatchのランチャーとして使っていただくために作ったプログラムです。

1. 以下のプログラムをF15JWatchの0番に保存します。
2. BTNを押したままF15JWatchの電源を入れると起動します。
3. IN1またはIN4のボタンでファイルを選択します。
4. 目的のプログラム番号が表示されたらBTNを押すと選択したプログラムが起動します。



●このランチャは保存されているプログラムの1行目の2文字目からあとの8文字分をLCDの2行目に表示するようにプログラムしています。1行目に8文字でもわかるプログラム名をつけることをお勧めします。

Special thanks: 笹野さん、後田さん（キャラクタの作成方法に関して）

```
10 'F15JWatch
20 PLAY "L32<<BRF"
30 CLS:CLP:CLV:X=1:A=#940:B=#700:C=B+1:D=62:E=3
40 POKE#7D0,4,6,15,20,10,20,10,20
50 POKEB,64,0,2,#C0,57,17,#70+E,86,#6C,56,12,1
60 R=I2CW(D,C,1,B+4,5):WAIT9:R=I2CW(D,C,1,B+9,3)
70 R=I2CW(D,C,1,B,1)+I2CW(D,B,1,#7D0,8)
80 ?"Welcome!":GSB370:WAIT60
90 POKE#920,70,0:LC2,1:?"JWatch":GSB380:WAIT90
100 LC0,2:?"
110 LC0,2:FILESX,X
120 FORI=1TO3:IFPEEK(A+I)=32GSB330
130 NEXT
140 K=265&IN(<)
150 IFK=257K=29
160 IFK=264K=28
170 IFK=9K=32
180 IFK=8K=31
190 IFK=1K=30
200 IFK<28ORK>32K=INKEY(<)
210 IFK<28ORK>32GOTO140
220 IFK=28X=X-1
230 IFK=29X=X+1
240 IFK=30E=3:GOTO50
250 IFK=31E=0:GOTO50
260 IFK=32LRUNX:END
270 IFX<1X=227
280 IFX=4X=100
290 IFX=99X=3
300 IFX>227X=1
310 GOTO100
320 'SCR
330 LC0,0:?"FILE#":LC5,0:?"STR$(A,I):GSB370
340 LC0,1:?"":LC0,1:?"STR$(A+I+2,8):GSB380
350 BEEP:I=9:RTN
360 'LCD
370 R=I2CW(D,C,1,B+2,1)+I2CW(D,B,1,#900,8):WAIT9:RTN
380 R=I2CW(D,C,1,B+3,1)+I2CW(D,B,1,#920,8):WAIT9:RTN
```


# プログラムの簡単な解説

```
10 'F15JWatch
20 PLAY"L32<<BRF" ← 起動音
30 CLS:CLP:CLV:X=1:A=#940:B=#700:C=B+1:D=62:E=3
```

↑  
画面、PCG、変数の初期化

X=ファイル番号  
A=LC0,2のVRAMアドレス  
B=PCGの先頭アドレス  
C=#701  
D=AQM0802AのI2Cスレーブアドレス (#7C>>1=62)  
E=コントラスト用 (#0-#Fまで)

```
40 POKE#7D0,4,6,15,20,10,20,10,20
```



5x8のイチゴのキャラクターを作成し、  
#7D0のPCG領域に記録  
(#7D0以外でもOK)

```
50 POKEB,64,0,2,#C0,57,17,#70+E,86,#6C,56,12,1
```

B(=#700)からのアドレスにLCDのコマンドを設定

#700: 64=データ:最終データ:Co=0、データ:RS=1(`01000000)  
#701: 0=コマンド:最終データ:Co=0、コマンド:RS=0(`00000000)  
#702: 2=ホームに戻る(LCDの1行目1文字目)  
#703: #C0=LCDの2行目1文字目  
#704: 57=機能設定(拡張モード)  
#705: 17=内部OSC 1/5base, 1  
#706: #70+E=コントラスト(#70-#7Fまで)  
#707: 86=アイコンOFF,ブースタON,コントラスト2  
#708: #6C=FollowerコントロールON, 4  
#709: 56=機能設定(ノーマルモードモード)  
#70A: 12=ディスプレイON  
#70B: 1=LCDの画面消去

```
60 R=I2CW<D,C,1,B+4,5>:WAIT9:R=I2CW<D,C,1,B+9,3>
```

LCDコマンドを実行

#704から5バイト分(拡張モード)

LCDコマンドを実行

#709から3バイト分(ノーマルモード)

```
70 R=I2CW<D,C,1,B,1>+I2CW<D,B,1,#7D0,8>
```

1、64(=CGRAMセットコマンド)を送信  
2、40行で作ったイチゴの文字をLCDのCGRAMの0文字目に転送  
●転送先アドレス=64+文字目×8バイト

```
80 ?"Welcome!":GSB370:WAIT60
```

```
90 POKE#920,70,0:LC2,1:?"JWatch":GSB380:WAIT90
```

80行: LC0,0からWelcome!とモニターに表示:GSB370(LCDに表示):1秒間待つ

90行: LC0,1のVRAM(#920)に70("F")と0(イチゴの文字)を表示

LC2,1からJwatchとモニターに表示

GSB380(LCDに表示):1秒半待つ

```
100 LC0,2:?" "
```

```
110 LC0,2:FILESX,X
```

```
120 FORI=1TO3:IFPEEK(A+I)=32GSB330
```

```
130 NEXT
```

100行: LC0,2をスペースで消す

110行: LC0,2にX番のファイルを表示

120行: LC0,2に表示された、行番号とプログラムの間のスペース(=32)をVRAM上で探す  
もし見つかったら330行を実行

```

140 K=265&IN(>)
150 IFK=257K=29
160 IFK=264K=28
170 IFK=9K=32
180 IFK=8K=31
190 IFK=1K=30
200 IFK<280RK>32K=INKEY(>)
210 IFK<280RK>32GOTO140
220 IFK=28X=X-1
230 IFK=29X=X+1
240 IFK=30E=3:GOTO50
250 IFK=31E=0:GOTO50
260 IFK=32LRUNX:END
270 IFX<1X=227
280 IFX=4X=100
290 IFX=99X=3
300 IFX>227X=1
310 GOTO100

```

\* ボタンの同時押しは困難なので、  
コントラストの調整は難しい

140行 IN1=1、IN4=8、BTN=256  
1+8+256=265でビット演算（アンド）  
150行 IN4を押した場合(265-8=257)、K=29  
160行 IN1を押した場合(265-1=264)、K=28  
170行 BTNを押した場合(265-256=9)、K=28  
180行 IN1+BTNを押した場合、K=31  
190行 IN4+BTNを押した場合、K=30  
200行 キーボード操作した場合の処理  
210行 Kが28から32の間でなければ140行へ（再取得）  
  
220行 ファイル番号を1減らす  
220行 ファイル番号を1増やす  
230行 コントラスト設定を#73にする：初期化へ  
240行 コントラスト設定を#70にする：初期化へ  
250行 プログラムスタート  
  
270行 ファイル番号が1より小さい場合、227番にする  
280行 ファイル番号が4の場合100番にする  
290行 ファイル番号が99の場合3番にする  
300行 ファイル番号が227より大きい場合1番にする  
  
310行 ファイル情報を取得・表示のため100行へ

```

320 ^SCR
330 LC0,0:? "FILE#      ":LC5,0:?STR$(A,I):GSB370
340 LC0,1:? "          ":LC0,1:?STR$(A+I+2,8):GSB380
350 BEEP:I=9:RTN

```

330行 LC0,0からFILE#と表示  
LC0,2のVRAMからファイル番号分をLC5,0にコピー  
VRAMの1行目をLCDの1行目に転送（GSB370）  
340行 LC0,1からスペースで消去  
LC0,2のVRAMからプログラム2文字目から8文字をLC0.1にコピー  
VRAMの2行目をLCDの2行目に転送（GSB380）

```

360 ^LCD
370 R=I2CW(D,C,1,B+2,1)+I2CW(D,B,1,#900,8):WAIT9:RTN
380 R=I2CW(D,C,1,B+3,1)+I2CW(D,B,1,#920,8):WAIT9:RTN

```

370行 LCDの1行目（#2=ホーム）に移動するコマンドを送信（0）  
VRAMの1行目（#900から）の8文字のデータをLCDに送信（64）  
380行 LCDの2行目（#C0）に移動するコマンドを送信（0）  
VRAMの2行目（#920から）の8文字のデータをLCDに送信（64）